AD-A068 208 OHIO STATE UNIV COLUMBUS ELECTROSCIENCE LAB F/G 17/9
THE CALCULATION OF FAR FIELD SCATTERING BY A CIRCULAR METALLIC --ETC(U)
FEB 79 D B HODGE N00014-78-C-0049
UNCLASSIFIED ESL-710816-2 NL

1 OF 1
ADA
068208



END
DATE
FILMED
6-79
DDC


1·0
2·8
2·5
3·15
2·2
3·5
2·0
1·1
4·0
4·5
1·8
1·25
1·4
1·6
NATIONAL BUREAU OF STANDARDS
MICROCOPY RESOLUTION TEST CHART

LEVEL II

OSU

THE CALCULATION OF FAR FIELD SCATTERING BY A CIRCULAR METALLIC DISK

D. B. Hodge

The Ohio State University
ElectroScience Laboratory
Department of Electrical Engineering
Columbus, Ohio 43212

TECHNICAL REPORT 710816-2

Contract N00014-78-C-0049

February 1979

Dept. of the Navy
Office of Naval Research
800 Quincy Street
Arlington, Virginia 22217

NOTICES

When Government drawings, specifications, or other data are used for any purpose other than in connection with a definitely related Government procurement operation, the United States Government thereby incurs no responsibility nor any obligation whatsoever, and the fact that the Government may have formulated, furnished, or in any way supplied the said drawings, specifications, or other data, is not to be regarded by implication or otherwise as in any manner licensing the holder or any other person or corporation, or conveying any rights or permission to manufacture, use, or sell any patented invention that may in any way be related thereto.

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

REPORT DOCUMENTATION PAGE

READ INSTRUCTIONS BEFORE COMPLETING FORM

1. REPORT NUMBER	2. GOVT ACCESSION NO.	3. RECIPIENT'S CATALOG NUMBER
4. TITLE (and Subtitle) (6) THE CALCULATION OF FAR FIELD SCATTERING BY A CIRCULAR METALLIC DISK.		5. TYPE OF REPORT & PERIOD COVERED (9) Technical Report.
		6. PERFORMING ORG. REPORT NUMBER (14) ESL-710816-2
7. AUTHOR(s) (10) D. B. Hodge		8. CONTRACT OR GRANT NUMBER(s) Contract N00014-78-C-0049 (15)
9. PERFORMING ORGANIZATION NAME AND ADDRESS The Ohio State University ElectroScience Laboratory, Department of Electrical Engineering Columbus, Ohio 43212		10. PROGRAM ELEMENT, PROJECT, TASK AREA & WORK UNIT NUMBERS Project NR 371-021/9-5-78 (427)
11. CONTROLLING OFFICE NAME AND ADDRESS Dept. of the Navy, Office of Naval Research, 800 Quincy Street Arlington, Virginia 22217	(11)	12. REPORT DATE Febr[redacted]79
		13. NUMBER OF PAGES 35
14. MONITORING AGENCY NAME & ADDRESS(if different from Controlling Office) (12) 40p.		15. SECURITY CLASS. (of this report) Unclassified
		15a. DECLASSIFICATION/DOWNGRADING SCHEDULE

16. DISTRIBUTION STATEMENT (of this Report)

This document has been approved for public release and sale; its distribution is unlimited.

17. DISTRIBUTION STATEMENT (of the abstract entered in Block 20, if different from Report)

18. SUPPLEMENTARY NOTES

19. KEY WORDS (Continue on reverse side if necessary and identify by block number)

Scattering
Radar cross section
Disk
Bistatic

20. ABSTRACT (Continue on reverse side if necessary and identify by block number)

A computer program for the general problem of far field scattering by a thin, circular, metallic disk is described. This program permits incident plane EM waves of arbitrary incidence direction and polarization and computes the amplitude and phase of both components of the scattered field at any point on the far field sphere. Thus, the program is applicable to any monostatic or bistatic far field scattering geometry.

over

DD FORM 1 JAN 73 1473 EDITION OF 1 NOV 65 IS OBSOLETE

UNCLASSIFIED

SECURITY CLASSIFICATION OF THIS PAGE (When Data Entered)

402 251

The program is based on Andrejewski's rigorous eigenfunction solution to the disk scattering problem. This report describes the solution and required spheroidal functions as well as the resulting program and its use.

TABLE OF CONTENTS

I. INTRODUCTION

At the present time the sphere represents the only radar target of finite extent for which numerical scattering results may be obtained directly and simply from the rigorous eigenfunction solution of the plane wave scattering problem. The rigorous eigenfunction solution of the thin metallic disk problem has been available in the literature for a considerable period of time [1]; however, only limited numerical results have become available due to the difficulty of the numerical computations required.

This state of affairs is quite unfortunate since the disk offers significant advantages as a standard radar calibration target when compared with the sphere. First, precision machining of a disk is much simpler than that of a sphere; and, second, precise alignment of the target for phase measurements is considerably simpler for a disk than for a sphere.

Furthermore, a great deal of basic understanding of scattering mechanisms is potentially available from the study of the thin disk. This is a consequence of the fact that it is the only target of finite extent, other than the sphere, for which a rigorous solution is available; and it is the only case for targets having a sharp edge. Thus, a complete understanding of scattering by a disk would provide another canonical solution to complement that of the sphere.

For these reasons, earlier work at The Ohio State University ElectroScience Laboratory [2,3,4] has been extended to generate a computer program capable of handling the general problem of far field scattering of a plane wave by a thin, metallic disk. This program permits incident plane waves of arbitrary incidence direction and arbitrary polarization and computes the amplitude and phase of both components of the scattered field at any point on the far field sphere.

The program has been generated in a user oriented form so that it can readily be used by any investigator without a detailed knowledge of the program. The program requires about 16K of core memory and executes in a matter of seconds on the ESL Datacraft 6024 computer operating in a time sharing mode.

II. THE GEOMETRY

The disk of radius a is centered at the origin and lies in the x-y plane. The direction of propagation of the incident plane wave is taken to be in the x-z plane without loss of generality. The angle of incidence, θ_0, is measured from the positive z-axis, i.e., the normal to the disk, as shown in Figure 1. The scattered far field is to be evaluated in a direction specified by the conventional spherical coordinates, θ_s and ϕ_s.

The polarization of the incident $\overline{E}$-field is aligned at an angle of α measured from the plane of incidence in the $+\phi$ direction. Thus, $\alpha=0$ is associated with the parallel, E-plane, or θ-polarized case, and $\alpha=\frac{\pi}{2}$ is associated with the perpendicular, H-plane, or ϕ-polarized case. Both the θ and ϕ components of the scattered $\overline{E}$-field will be determined.

III. THE EIGENFUNCTION SOLUTION

The solution presented here parallels that obtained by Andrejewski [1]. For convenience in the computation, the solution has been cast in terms of trigonometric rather than exponential functions. And, the more conventional notation of Flammer [5] has been followed.

The incident electric field intensity is given by

$$\overline{E}^i = E_0\,(-\cos\theta_0\cos\alpha\hat{a}_x + \sin\alpha\hat{a}_y + \sin\theta_0\cos\alpha\hat{a}_z)e^{i(\overline{k}^i\cdot\overline{r}+\omega t)} \qquad (1)$$



Figure 1. The geometry of the scattering problem.

where

$$\overline{k}^i = k\,\hat{k}^i$$

$$k = \omega\sqrt{\mu_o \epsilon_o} \tag{3}$$

$$\hat{k}^i = -\sin\theta_o \hat{a}_x - \cos\theta_o \hat{a}_z . \tag{4}$$

Then, the scattered electric field intensity at an arbitrary point, (r,θ_s,ϕ_s), on the far field sphere may be expressed as

$$\overline{E}^s = \frac{iE_o}{kr}\left(\frac{\cos\alpha}{\cos\theta_o}\,\overline{e}_{\parallel} + \sin\alpha\,\overline{e}_{\perp}\right)\overline{e}^{i(kr-\omega t)} . \tag{5}$$

It is convenient to express the scattered field $\overline{E}^s$ in terms of a normalized field $\overline{E}_n^s$ as

$$\overline{E}^s = \frac{aE_o}{2r}\,\overline{E}_n^s\, e^{-i(kr-\omega t)} . \tag{6}$$

This choice of normalization yields a particularly simple form for the normalized radar cross section, i.e.,

$$\frac{\sigma}{\pi a^2} = |\overline{E}_n^s|^2 . \tag{7}$$

The normalized electric field intensity in this case is given by

$$\overline{E}_n^s = \frac{2i}{ka}\left(\frac{\cos\alpha}{\cos\theta_o}\,\overline{e}_{\parallel} + \sin\alpha\overline{e}_{\perp}\right) \tag{8}$$

where

$$e_{\parallel_\phi} = \cos\theta \sum_{m=0}^{\infty} \left\{ -2(2-\delta_{o,m})\cos(m\phi)\cos\phi \cdot Y_m(\cos\theta,c,\cos\theta_o) \right.$$

$$\left. + i^{-m}\left[U_{m+1}\cos(m+1)\phi-(1+\delta_{m,1})U_{m-1}\cos(m-1)\phi\right] Y_m(\cos\theta,c,0)\right\}$$

$$e_{\parallel_\phi} = \sum_{m=0}^{\infty} \left\{ -2(2-\delta_{o,m})\cos(m\phi)\sin\phi \cdot Y_m(\cos\theta,c,\cos\theta_o) \right.$$

$$\left. + i^{-m}\left[U_{m+1}\sin(m+1)\phi+U_{m-1}\sin(m-1)\phi\right] Y_m(\cos\theta,c,0)\right\}$$

$$e_{\perp_\theta} = \cos\theta \sum_{m=0}^{\infty} \left\{ 2(2-\delta_{o,m})\cos(m\phi)\sin\phi \cdot Y_m(\cos\theta,c,\cos\theta_o) \right.$$

$$\left. - i^{-m}\left[X_{m+1}\sin(m+1)\phi-X_{m-1}\sin(m-1)\phi\right] \overline{Y}_m(\cos\theta,c,0)\right\}$$

$$e_{\perp_\phi} = \sum_{m=0}^{\infty} \left\{ -2(2-\delta_{o,m})\cos(m\phi)\cos\phi \cdot Y_m(\cos\theta,c,\cos\theta_o) \right.$$

$$\left. + i^{-m}\left[X_{m+1}\cos(m+1)\phi+(1-\delta_{m,1})X_{m-1}\cos(m-1)\phi\right] Y_m(\cos\theta,c,0)\right\} \tag{9}$$

and

$$c = ka \quad . \tag{10}$$

The functions Y_m are given in terms of the spheroidal radial functions, $R_{mn}^{(i)}(-ic;i0)$, and the spheroidal angular functions, $S_{mn}(-ic,\cos\theta)$, by

$$Y_m(\cos\theta,c,\cos\theta_o) = \sum_{\substack{n=m \\ n-m \text{ even}}}^{\infty} \frac{(-1)^n}{N_{mn}(-ic)} \frac{R_{mn}^{(1)}(-ic;io)}{R_{mn}^{(4)}(-ic;io)} S_{mn}(-ic,\cos\theta_o)S_{mn}(-ic;\cos\theta) \tag{11}$$

The prime on the summation symbol emphasizes the fact that the summation over n proceeds by increments of 2 as a consequence of the condition that n-m is even. The normalization function, N_{mn}, and the spheroidal functions will be described later.

The U and X functions are given by

$$U_m = 2i^{m-1} \frac{W_{m-1} + W_{m+1}}{\psi_{m-1} + \psi_{m+1}} , \quad m \geq 1$$

$$U_o = - i \frac{W_1}{\omega_1} \tag{12}$$

$$U_m = 0, \quad m < 0$$

and

$$X_m = 2i^{m-1} \frac{W_{m-1} - W_{m+1}}{\psi_{m-1} + \psi_{m+1}} , \quad m \geq 1$$

$$X_m = 0, \quad m \leq 0 \tag{13}$$

Finally, the W and ψ functions are given in terms of the spheroidal functions by

$$W_m = \sum_{\substack{n=m \\ n-m \text{ even}}}^{\infty}{}' \frac{i^n}{N_{mn}(-ic)} \cdot \frac{S_{mn}(-ic,\cos\theta_o)S_{mn}(-ic;o)}{R_{mn}^{(4)}(-ic;io)} \tag{14}$$

and

$$\psi_m = \sum_{\substack{n=m \\ n-m \text{ even}}}^{\infty}{}' \frac{i^n}{N_{mn}(-ic)} \cdot \frac{\left[S_{mn}(-ic,o)\right]^2}{R_{mn}^{(4)}(-ic;io)} \quad . \tag{15}$$

The angular spheroidal functions may be expressed in terms of conventional spherical Legendre polynomials, $P^m_{m+r}(\cos\theta)$, as [5]

$$S_{mn}(-ic;\cos\theta) = \sum_{r=0,1}^{\infty}{}' \; d_r^{mn}(-ic) P^m_{m+r}(\cos\theta) \tag{16}$$

where the prime indicates that the summation is over even values of r if n-m is even and over odd values of r if n-m is odd. The expansion coefficients, $d_r^{mn}(-ic)$, are those used by Flammer [5] and may be computed readily using a technique described in Reference 2.

Since only spheroidal radial functions of zero argument and n-m even are required, the special relationships for these cases as presented by Flammer may be used:

$$R^{(1)}_{mn}(-ic;io) = \frac{i^{n-m} 2^m m!\, c^m d_0^{mn}(-ic)}{(2m+1) \sum_{r=0}^{\infty}{}' \; d_r^{mn}(-ic) \frac{(2m+r)!}{r!}} \tag{17}$$

$$R^{(2)}_{mn}(-ic;io) = \frac{i^{n-m}(2m-1) m!\, c^{m-1} \pi}{2^{2n-m+1}(2m)!\, d_{-2m}^{mn}(-ic) \sum_{r=0}^{\infty}{}' \; d_r^{mn}(-ic) \frac{(2m+r)!}{r!}} \cdot \left[\frac{(n+m)!}{\left(\frac{n-m}{2}\right)! \left(\frac{n+m}{2}\right)!} \right]^2 . \tag{18}$$

The expansion coefficients, $d_r^{mn}(-ic)$, used here are identical to those used in Equation (16). The radial spheroidal functions of the 4^{th} kind are found simply as in the spherical case by

$$R^{(4)}(-ic;io) = R^{(1)}(-ic;io) - iR^{(2)}(ic;io). \tag{19}$$

The normalization functions, $N_{mn}(-ic)$, are those required to cause the angular spheroidal functions to satisfy

$$S_{mn}(-ic,o) = P_n^m(o) \qquad (20)$$

and are given by

$$N_{mn}(-ic) = 2 \sum_{r=0,1}^{\infty\,'} \frac{(r+2m)!\left[d_r^{mn}(-ic)\right]^2}{(2r+2m+1)r!} \quad . \qquad (21)$$

Equations (8) through (21) provide the complete solution to the general far field scattering problem.

IV. THE PROGRAM

Using the solution given in the preceeding section, a Fortran computer program was prepared for the calculation of far field scattering by a circular metallic disk. The program was developed on a time-sharing Datacraft 6024 machine having a 24 bit word length. The storage requirements are

Main Program and Subroutines	$(4{,}790)_{10}$ words
Library Subroutines	$(4{,}514)_{10}$ words
Common Storage	$(7{,}300)_{10}$ words
Total Storage	$(16{,}604)_{10}$ words

(not including operating system and I/O buffers).

This program executes a complete far field scattering computation in a matter of seconds in the time-sharing environment.

A sample of the program I/O is shown in Figure 2. In general, one need only provide 8 variables for the initial case to be executed; they are:

1. KA = the electrical radius of the disk $= \frac{2\pi a}{\lambda}$
2. THETA INCIDENT = θ_0 [degrees] (see Figure 1)
3. POLARIZATION = α [degrees]
4. THETA SCATTERED = θ_s [degrees]
5. PHI SCATTERED = ϕ_s [degrees]

WHICH VARIABLE IS TO BE INCREMENTED?

if 1 thru 5:	then the variable associated with that index above will be incremented
if not 1 thru 5:	then only the initial case will be calculated. In this event the remaining parameters are not requested.

TYPE NUMBER OF CASES; enter the total number of computations to be performed.

WHAT IS THE INCREMENT: enter the increment by which the variable selected should be increased after each execution.

All inputs may be made in free format.

The program labels the 1st column with the name of the variable to be incremented. The 2nd and 3rd columns contain the cross sections (Equation (7)) associated with the θ and ϕ components of the scattered field. The last four columns list the magnitudes and phases (in degrees) of both the θ and ϕ components of the normalized scattered electric field, $\overline{E}_n^s$ (Equation (8)). It should be noted that all of the elements of the scattering matrix are available in lines 95-104 of the program.

```
SCATTERING BY A METALLIC CIRCULAR DISK
     (HODGE -- VERSION  12/17/78)


(TYPE "ESC" TO RESTART PROGRAM)
(TYPE KA=0 TO STOP PROGRAM)
(TYPE KA=-1 FOR A DESCRIPTION OF THE PARAMETERS)
(NORMALIZATION:   ESCAT=A*EINC*ENORM/(2*R)*EXP(-J*K*R)
(ALL ANGLES IN DEGREES)


1.  KA                =  1

2.  THETA INCIDENT    =  0

3.  POLARIZATION      =  0

4.  THETA SCATTERED   =  0

5.  PHI SCATTERED     =  0


WHICH VARIABLE IS TO BE INCREMENTED?  1

TYPE NUMBER OF CASES:   29

WHAT IS THE INCREMENT? .5
```

KA	CROSS SECTION SIGMA/(PI*A**2) THETA		CROSS SECTION SIGMA/(PI*A**2) PHI		E NORM THETA MAG		E NORM THETA PHASE	E NORM PHI MAG		E NORM PHI PHASE
1.00	.183E	1	.000E	1	.135E	1	-21.98	.000E	1	90.00
1.50	.772E	1	.000E	1	.278E	1	-65.64	.000E	1	90.00
2.00	.907E	1	.000E	1	.301E	1	-86.77	.000E	1	90.00
2.50	.101E	2	.000E	1	.318E	1	-93.01	.000E	1	90.00
3.00	.115E	2	.000E	1	.339E	1	-94.66	.000E	1	90.00
3.50	.132E	2	.000E	1	.363E	1	-94.00	.000E	1	90.00
4.00	.155E	2	.000E	1	.393E	1	-91.37	.000E	1	90.00
4.50	.197E	2	.000E	1	.443E	1	-87.71	.000E	1	90.00
5.00	.271E	2	.000E	1	.521E	1	-86.78	.000E	1	90.00
5.50	.346E	2	.000E	1	.583E	1	-89.14	.000E	1	90.00
6.00	.398E	2	.000E	1	.631E	1	-91.13	.000E	1	90.00
6.50	.440E	2	.000E	1	.664E	1	-91.73	.000E	1	90.00
7.00	.485E	2	.000E	1	.696E	1	-91.13	.000E	1	90.00
7.50	.547E	2	.000E	1	.739E	1	-89.67	.000E	1	90.00
8.00	.645E	2	.000E	1	.803E	1	-88.49	.000E	1	90.00
8.50	.763E	2	.000E	1	.874E	1	-89.02	.000E	1	90.00
9.00	.859E	2	.000E	1	.927E	1	-90.25	.000E	1	90.00
9.50	.931E	2	.000E	1	.965E	1	-90.92	.000E	1	90.00
10.00	.100E	3	.000E	1	.100E	2	-90.84	.000E	1	90.00
10.50	.108E	3	.000E	1	.104E	2	-90.12	.000E	1	90.00
11.00	.120E	3	.000E	1	.110E	2	-89.05	.000E	1	90.00
11.50	.135E	3	.000E	1	.116E	2	-89.13	.000E	1	90.00
12.00	.150E	3	.000E	1	.122E	2	-89.90	.000E	1	90.00
12.50	.160E	3	.000E	1	.127E	2	-90.51	.000E	1	90.00
13.00	.170E	3	.000E	1	.130E	2	-90.63	.000E	1	90.00
13.50	.180E	3	.000E	1	.134E	2	-90.27	.000E	1	90.00
14.00	.194E	3	.000E	1	.139E	2	-89.65	.000E	1	90.00
14.50	.212E	3	.000E	1	.146E	2	-89.37	.000E	1	90.00
15.00	.231E	3	.000E	1	.152E	2	-89.75	.000E	1	90.00

Figure 2. I/O listing.

Upon completion of the cases desired, the program will request a new disk size and proceed as before. At any time one may type "ESC" and cause the current task to be terminated; the program will then again request a new disk size and proceed as before.

Entering a disk size of 0 will cause the program to terminate. Entering a disk size of -1 will cause a brief statement of the problem geometry to be printed; following this a new disk size will be requested.

A simplified flow diagram showing the computational logic is presented in Figure 3. The logic is quite straight forward and proceeds along the line specified by Equations (7-21).

It is necessry to compute the eigenvalues of the spheroidal wave equation, $\lambda_{mn}(-ic)$, in order to determine the expansion coefficients $d^r_{mn}(-ic)$, appearing in Equations (16), (17), (18) and (20). The eigenvalues are computed by the bisection method and the expansion coefficients are computed by recursion as described in Reference 2.

The solution of the scattering problem consists of a triple summation over the indices m, n, and r. The truncation of these summations is performed internally by the software. Various functions are examined to determine if they are near the machine overflow level, i.e., 10^{38} for the Datacraft 6024. If this level is reached, the appropriate summation is truncated as described in Appendix B. This procedure yields the best possible convergence for a machine having this dynamic range. This procedure has also been successfully used for sphere scattering calculations. In both cases the truncation is controlled largely by the tendency of the radial functions appearing in denominators to become extremely large. This tends to insure adequate convergence of the solution.



Figure 3. Simplified flow diagram. (The numbers in parentheses refer to line numbers in the program listing presented in Appendix I).

V. SAMPLE RESULTS

Some sample results are included here to serve as check cases and to demonstrate the utility of the program.

Calculations were made as a function of electrical disk size for normal incidence backscatter. These results are tabulated in Figure 2; and the normalized radar cross section and scattered E-field phase are plotted in Figures 4 and 5, respectively. The Rayleigh or low frequency approximation

$$E^s_{n_{Ray}} = \frac{8(ka)^2}{3\pi}$$

and the Geometrical Theory of Diffraction or high frequency approximation

$$E^s_{n_{GTD}} = \frac{1}{\sqrt{\pi ka}}\, e^{-i(2ka + \frac{3\pi}{4})} - \frac{3i}{4ka} + \frac{1}{2\pi ka}\, e^{-i(4ka - \frac{\pi}{2})} - ika$$

are also shown presented in these figures for comparison.

Results for normal incidence, bistatic scattering are shown in Figures 6 and 7 as a function of the scattering direction. Both $\bar{E}$- and $\bar{H}$-plane results are given here for a disk size of ka=10 (diameter = 3.18λ).

Results for specular bistatic $\bar{E}$- and $\bar{H}$-plane scattering from a disk of ka=10 are shown in Figures 8 and 9. In this case $\theta_i=\theta_s$ and $\phi_s = 180^o$. Note that the phases for $\theta_s=0$ in Figures 7 and 9 differ by 180^o; this is a consequence of the chosen coordinate system. The calculations in Figure 7 were done for $\phi_s=0$ and those in Figure 9 for $\phi_s=180^o$.



Figure 4. Normal incidence, backscatter.



Figure 5. Normal incidence, backscatter.



Figure 6. Normal incidence, bistatic scatter. (ka=10).



Figure 7. Normal incidence, bistatic scatter. (ka=10).



Figure 8. Bistatic specular scatter. (ka=10)



Figure 9. Bistatic specular scatter. (ka=10)

VI. SUMMARY

The expressions for calculating far field scattering characteristics of a plane wave by a thin metallic circular disk were presented. These expressions are applicable for any incident polarization or direction of incidence and for both scattered polarizations at any point on the far field sphere. This formulation was used to develop a user oriented computer program which is also described herein. This program has been used successfully for disk sizes up to ka=15. It yields both the radar cross section and phase of both scattered field components. The program executes in a matter of seconds and requires about 16k of core memory.

REFERENCES

1. "Die Beugung Elektromagnetischer Wellen an der Leitenden Kreisscheibe und an der Kreisformigen Offnung im Leitenden Ebenen Schieum," W. Andrejewski, Dissertation for Rheinich-Westfalichen Technischen Hochschule, Aachen, Germany, 1952.

2. D. B. Hodge, "Eigenvalues and Eigenfunctions of the Spheroidal Wave Equation," J. Mathematical Physics, Vol. 11, p. 2308, 1970.

3. D. B. Hodge, "Spectral and Transient Response of a Circular Disk to Plane Electromagnetic Waves," IEEE Trans. Vol. AP-19, p. 558, 1971.

4. R. DeVore, D. B. Hodge and R. G. Kouyoumjian, "Backscattering Cross Sections of Circular Disks of Arbitrary Incidence," J. Appl. Phys., Vol. 42, p. 3075, 1971.

5. C. Flammer, "Spheroidal Wave Functions," Stanford University Press, 1957.

APPENDIX I
PROGRAM LISTING

The following is a Fortran listing of the program described in the body of this report. The use of this program is described in the section entitled the Program. Comments on the program are included in Appendix II.

```
C        FAR FIELD SCATTERING BY A CIRCULAR METALLIC DISK
         INCLUDE MESSB,SYS9
         DIMENSION LABEL(12),VAR(5)
         COMPLEX F4,PSI,W,YO,YETAO,U,X,Z,ZA,ZB,ZC,ZD
        1,IX,EPARI,EPARP,EPERI,EPERP,ESNT,ESNP
         COMMON EIG(50),D(50,50),DDEG(50),R1(50),F4(50)
        1,SO(50),R(50),SETA(50),SETAO(50),PSI(50),W(50),YO(50)
        1,YETAO(50),U(50),X(50),CMPHI(50),SMPHI(50)
         DATA LABEL/36H    KA THE I    POL THE S PHI S         /
         IX=(0.,1.)
         WRITE(8,11)
 11      FORMAT(1X,///,1X,'SCATTERING BY A METALLIC CIRCULAR DISK',
        1/,5X,'(HODGE -- VERSION  12/17/78)')
         WRITE(8,26)
 26      FORMAT(1X,///,'(TYPE "ESC" TO RESTART PROGRAM)',/,
        1'(TYPE KA=0 TO STOP PROGRAM)',/,
        1'(TYPE KA=-1 FOR A DESCRIPTION OF THE',
        1' PARAMETERS)',/,'(NORMALIZATION:  ESCAT=A*EINC*ENORM'
        1'/(2*R)*EXP(-J*K*R))',/,'(ALL ANGLES IN DEGREES)')
         CALL ESC($42)
 4       NOC=1
         ILL1=11
         INDEX=1
         WRITE(8,27)
 27      FORMAT(1X,///,1X,'1.  KA',14X,'= ')
         READ(8,-) VAR(1)
         IF(VAR(1).EQ.-1)GO TO 41
         IF(VAR(1).LE.0.)GO TO 5
         WRITE(8,28)
 28      FORMAT(1X,'2.  THETA INCIDENT  = ')
         READ(8,-) VAR(2)
         WRITE(8,29)
 29      FORMAT(1X,'3.  POLARIZATION    = ')
         READ(8,-) VAR(3)
         WRITE(8,30)
 30      FORMAT(1X,'4.  THETA SCATTERED = ')
         READ(8,-) VAR(4)
         WRITE(8,31)
 31      FORMAT(1X,'5.  PHI SCATTERED   = ')
         READ(8,-) VAR(5)
         WRITE(8,3)
 3       FORMAT(1X,//,1X,'WHICH VARIABLE IS TO BE INCREMENTED?')
         READ(8,-)NVAR
         IF((NVAR.LE.0).OR.(NVAR.GT.5))GO TO 23
         WRITE(8,21)
 21      FORMAT(1X,'TYPE NUMBER OF CASES:')
         READ(8,-) NOC
         WRITE(8,32)
 32      FORMAT(1X,'WHAT IS THE INCREMENT?')
         READ(8,-)VINCRE
         ILL1=2*NVAR-1
 23      ILL2=ILL1+1
         WRITE(8,24)LABEL(ILL1),LABEL(ILL2)
 24      FORMAT(1X,//,2X,2A3,6X,'CROSS SECTION',21X,'E NORM',/,
        113X,'SIGMA/(PI*A**2)',11X,'THETA',15X,'PHI',/,13X,
```

```
     1'THETA',7X,'PHI',8X,'MAG',5X,'PHASE',6X,'MAG',5X,
     1'PHASE',/)
6     C=VAR(1)
      MMMAX=45
      NNMAX=45
      IRRMAX=45
      THE0=VAR(2)*3.14159/180
      ETA0=COS(THE0)
      THE=VAR(4)*3.14159/180
      ETA=COS(THE)
      PHI=VAR(5)*3.14159/180
      ALF=VAR(3)*3.14159/180
      CALF=COS(ALF)
      SALF=SIN(ALF)
34    DO 10 MM=1,MMMAX
      M=MM-1
      IQ=0
      CALL SEND(M,NNMAX)
      CALL OBEIGN(C,M,NNMAX)
      CALL OBCOEF(C,M,MMMAX,NNMAX,IRRMAX)
      CALL OBEGN(C,M,NNMAX)
      CALL OBRAD(C,M,IQ,MMMAX,NNMAX,IRRMAX)
      IF(IQ.EQ.1)GO TO 10
      CALL EPSI(M,NNMAX)
      CALL POLYN(ETA0,M,IRRMAX)
      CALL OBANG(NNMAX,IRRMAX)
      DO 7 I=1,NNMAX
7     SETA0(I)=SETA(I)
      CALL EX(M,NNMAX)
      CALL POLYN(ETA,M,IRRMAX)
      CALL OBANG(NNMAX,IRRMAX)
      CALL EY(M,NNMAX)
10    CONTINUE
      DO 13 MM=1,MMMAX
      CALL EXU(MM,MMMAX)
      IF(MMMAX.LT.MM)GO TO 13
      CALL CSPHI(MM,PHI)
13    CONTINUE
      CALL EZ(MMMAX,Z,ZA,ZB,ZC,ZD)
      EPART=ETA*(-2*Z*CMPHI(2)+ZA)
      EPARP=-2*Z*SMPHI(2)+ZB
      EPERT=ETA*(2*Z*SMPHI(2)-ZC)
      EPERP=-2*Z*CMPHI(2)+ZD
      IF(ETA0.EQ.0.) GO TO 2
      ESNT=2*IX*(CALF*EPART/ETA0+SALF*EPERT)/C
      ESNP=2*IX*(CALF*EPARP/ETA0+SALF*EPERP)/C
      GO TO 8
2     ESNT=2*IX*SALF*EPERT/C
      ESNP=2*IX*SALF*EPERP/C
8     EMAGT=CABS(ESNT)
      EMAGP=CABS(ESNP)
      SIGTHE=EMAGT*EMAGT
      SIGPHI=EMAGP*EMAGP
      E1=REAL(ESNT)
      E2=AIMAG(ESNT)
```

```
      IF(E1.EQ.0.) GO TO 16
      ARG=E2/E1
      EPHAT=180/3.14159*ATAN(ARG)
      IF((E1.LT.0.).AND.(E2.GT.0.))EPHAT=EPHAT+180
      IF((E1.LT.0.).AND.(E2.LT.0.))EPHAT=EPHAT-180
      GO TO 17
16    EPHAT=90
      IF(E2.LT.0.)EPHAT=-90
17    E1=REAL(ESNP)
      E2=AIMAG(ESNP)
      IF(E1.EQ.0.) GO TO 18
      ARG=E2/E1
      EPHAP=180/3.14159*ATAN(ARG)
      IF((E1.LT.0.).AND.(E2.GT.0.))EPHAP=EPHAP+180
      IF((E1.LT.0.).AND.(E2.LT.0.))EPHAP=EPHAP-180
      GO TO 19
18    EPHAP=90
      IF(E2.LT.0.)EPHAP=-90
19    IF(NVAR.EQ.0)NVAR=1
      WRITE(6,25)VAR(NVAR),SIGTHE,SIGPHI,EMAGT,EPHAT,
     1EMAGP,EPHAP
25    FORMAT(1X,F7.2,2(1X,E10.3),2(1X,E10.3,1X,F7.2))
27    IF(INDEX.EQ.NOC)GO TO 4
      INDEX=INDEX+1
      VAR(NVAR)=VAR(NVAR)+VINCRE
      GO TO 6
5     CALL EXIT
41    WRITE(6,40)
40    FORMAT(1X,//,'THE DISK OF RADIUS A LIES IN THE X-Y PLANE',/,
     1'CENTERED AT THE ORIGIN.  THE CONVENTIONAL (R, THETA,',/,
     1'PHI) COORDINATE SYSTEM IS USED IN THE FAR FIELD.  ',/,
     1'THE PLANE OF INCIDENCE OF THE PLANE WAVE IS THE',/,
     1'X-Z (PHI=0) PLANE.  THE POLARIZATION ANGLE (POL)',/,
     1'OF EINC IS MEASURED FROM THE PLANE OF INCIDENCE ',/,
     1'IN THE PHI-DIRECTION, I.E., POL=0 IS THE PARALLEL',/,
     1'(THETA) CASE AND POL=90 IS THE PERPENDICULAR (PHI)',/,
     1'CASE.',/,'THE RESULT IS A SOLUTION OF THE RIGOROUS',/,
     1'EIGENFUNCTION SCATTERING PROBLEM.')
      GO TO 4
42    CONTINUE
      GO TO 4
      END
C
C     OBLATE SPHEROIDAL ANGULAR FUNCTION, S, OF ARGUMENT 0;
C     ORDER M,N; WITH N-M EVEN; UP TO ORDER N=M+2*NNMAX-2.
C     (EQUAL TO PMN(0))
C
      SUBROUTINE SMNO(M,NNMAX)
      COMPLEX F4
      COMMON EIG(50),D(50,50),DNEG(50),R1(50),F4(50)
     1,S0(50)
      S0(1)=1
      IF(M.EQ.0)GO TO 1
      DO 2 MM=1,M
2     S0(1)=(2*MM-1)*S0(1)
```

```
1     DO 3 NN=1,NNMAX
      N=2*(NN-1)+M
3     S0(NN+1)=-(N+M+1)*S0(NN)/(N-M+2)
      RETURN
      END
C
C     SIN AND COS FUNCTIONS OF PHI
C
      SUBROUTINE CSPHI(MM,PHI)
      COMPLEX F4,PSI,W,Y0,YETA0,U,X
      COMMON EIG(50),D(50,50),DNEG(50),R1(50),F4(50)
     1,S0(50),P(50),SETA(50),SETA0(50),PSI(50),W(50),Y0(50)
     1,YETA0(50),U(50),X(50),CMPHI(50),SMPHI(50)
      M=MM-1
      CMPHI(MM)=COS(M*PHI)
      SMPHI(MM)=SIN(M*PHI)
      RETURN
      END
C
C     OBLATE SPHEROIDAL EIGENVALUES OF ARGUMENT C, ORDER M,N
C     WITH N-M EVEN UP TO ORDER N=M+2*NNMAX-2
C
      SUBROUTINE OBEIGN(C,M,NNMAX)
      COMMON EIG(50)
      DIMENSION IP(50),P(50),ALPHA(50),BETA(50)
4     CONTINUE
      M2=2*M
      C2=C*C
      ACC=1.0E-05
      NN2=NNMAX+2
      N1=NN2+1
      P(1)=1
      IP(1)=1
      DO 2 IQQ=1,NN2
      IV=2*IQQ-1
      IW=M2+2*IQQ
      IX=M2+4*IQQ-1
      ALPHA(IQQ)=(C2*(M2*(2*IV-1)+2*IV*(IV-1)-1))/(IX*(IX-4))
     1-(M+IV-1)*(IV+M)
2     BETA(IQQ+1)=C2/IX*SQRT(IV*(IV+1)*IW*(IW-1)/(IX*IX-4.0))
      BETA(NN2+1)=0.
      B0=ABS(ALPHA(1))+ABS(BETA(2))
      DO 3 IQQ=2,NN2
      A0=ABS(BETA(IQQ))+ABS(ALPHA(IQQ))+ABS(BETA(IQQ+1))
      BETA(IQQ)=BETA(IQQ)*BETA(IQQ)
      IF(A0.GT.B0) B0=A0
3     CONTINUE
      A0=-B0
      B01=B0
13    CONTINUE
      B0=B01
      DO 20 IQQ=1,NNMAX
      N=2*IQQ-2+M
      A=A0
      B=B0
```

```
      IERR=-1
   21 IIS=0
      C0=(A+B)/2
      IF(C0)50,22,50
   50 ERR=(B-A)/ABS(C0)
      IERR=IERR+1
      IF(IERR-60)40,41,41
   41 WRITE(6,42)N
   42 FORMAT(1X,'ITERATIONS EXCEEDED FOR EIGENVALUE ',I3)
      GO TO 700
   40 IF(ERR-ACC)24,24,22
   22 P(2)=ALPHA(1)-C0
      DO 5 I=3,NI
      P(I)=(ALPHA(I-1)-C0-BETA(I-1)*(P(I-2)/P(I-1)))*P(I-1)
      PMAG=ABS(P(I))
      IF(PMAG.GT.1.0E+33)GO TO 7
    5 CONTINUE
   12 CONTINUE
      DO 6 I=2,NI
      IF(P(I))14,8,9
    8 IF(P(I-1))9,9,14
   14 IP(I)=-1
      GO TO 10
    9 IP(I)=1
   10 IF(IP(I)-IP(I-1))6,11,6
   11 IIS=IIS+1
    6 CONTINUE
      IF(IIS-IGG)16,15,15
   15 A=C0
      GO TO 21
   16 B=C0
      GO TO 21
   24 B0=C0
  700 EIG(IGG)=-C0
   20 CONTINUE
      RETURN
    7 NNMAX=I-4
      NI=NNMAX+3
      GO TO 4
      END
C
C     OBLATE SPHEROIDAL EIGENFUNCTION EXPANSION COEFFICIENTS
C     OF ARGUMENT C: ORDER M,N,R: WITH N-M EVEN, UP TO ORDER
C     N=M+2*NNMAX-2, R=2*IRRMAX+2
C
      SUBROUTINE OBCOEN(C,M,MMMAX,NNMAX,IRRMAX)
      COMMON EIG(50),D(50,50)
      DIMENSION DP(50)
      C2=C*C
      MM=M+1
      DO 1 NN=1,NNMAX
      N=M+2*NN-2
    4 DP(IRRMAX+3)=0
      DP(IRRMAX+2)=1.0E-30
      D(NN,1)=0
```

```
      D(NN,2)=1
      JJ=(N-M)/2+1
      DO 107 LL=1,IRRMAX
      L=LL-1
      IF(LL.GE.JJ) L=IRRMAX+JJ-LL
      IR=2*L
      IRM=M+IR
      AR=(M+IRM+2)*(M+IRM+1)*C2/((2*IRM+3)*(2*IRM+5))
      BR=(2*IRM*(IRM+1)-2*M*M-1)*C2/((2*IRM-1)*
     1   (2*IRM+3))-IRM*(IRM+1)
      CR=IR*(IR-1)*C2/((2*IRM-3)*(2*IRM-1))
      IF(LL-JJ)105,106,106
  105 D(NN,L+3)=-(CR*D(NN,L+1)+(BR+EIG(NN))*D(NN,L+2))/AR
      DMAG=ABS(D(NN,L+3))
      IF(DMAG.GT.1.0E+30)GO TO 3
      GO TO 107
  106 DP(L+1)=-(AR*DP(L+3)+(BR+EIG(NN))*DP(L+2))/CR
      DMAG=ABS(DP(L+1))
      IF(DMAG.GT.1.0E+30)GO TO 3
  107 CONTINUE
      DL=ABS(D(NN,JJ+1))
      DL=ALOG10(DL)
      DLP=ABS(DP(JJ+1))
      DLP=ALOG10(DLP)
      DL=ABS(DL)
      DLP=ABS(DLP)
      DL=DL+DLP
      IF(DL.GT.30.)GO TO 5
      CON=D(NN,JJ+1)/DP(JJ+1)
      ACON=ABS(CON)
      IF(ACON.GE.1.0E+32)GO TO 2
      DO 118 J=JJ,IRRMAX
  118 D(NN,J+2)=CON*DP(J+2)
      F=1
      IF(M)198,198,199
 199  DO 110 I=1,M
 110  F=F*(M+I)
  198 SUM=0
      MMX=IRRMAX+1
      DO 113 I=1,MMX
      IR=2*I
      SUM=SUM+F*D(NN,I+1)
      IF(I-JJ) 113,197,113
  197 FNM=F
  113 F=(-F*(IR+2*M-1))/IR
      ALF=FNM/SUM
      DO 114 I=1,MMX
      D(NN,I)=ALF*D(NN,I+1)
 114  CONTINUE
    1 CONTINUE
      RETURN
 3    IRRMAX=LL-1
      GO TO 4
 2    IRRMAX=IRRMAX-1
      GO TO 4
```

```
5     NNMAX=NN-1
      RETURN
      END
C
C     NEGATIVE D COEFFICIENT SUBROUTINE
C
      SUBROUTINE DNEGN(C,M,NNMAX)
      COMMON EIG(50),D(50,50),DNEG(50)
      DO 4 NN=1,NNMAX
      C2=C*C
      IF (M.GE.1) GO TO 2
      DO 5 NN=1,NNMAX
5     DNEG(NN)=D(NN,1)
      GO TO 3
    2 B1=1.0
      B2=0.0
      EI=EIG(NN)
      DO 1 IRR=1,M
      IR=2*IRR-2*M-2
      AR=(2*M+IR+2)*(2*M+IR+1)*C2/((2*M+2*IR+3)
     1*(2*M+2*IR+5))
      BR=(M+IR)*(M+IR+1)-EI-(2*(M+IR)*(M+IR+1)-2*M*M-1)
     1*C2/((2*M+2*IR-1)*(2*M+2*IR+3))
      CR=(IR)*(IR-1)*C2/((2*M+2*IR-3)*(2*M+2*IR-1))
      B3=B2
      B2=B1
    1 B1=(BR*B2-CR*B3)/AR
      A=D(NN,1)/B1
      DNEG(NN)=A
      DDM=ABS(DNEG(NN))
      IF(DDM.LT.1.0E-35)GO TO 6
4     CONTINUE
    3 RETURN
6     NNMAX=NN-1
      RETURN
      END
C
C     OBLATE SPHEROIDAL RADIAL FUNCTION R(4) OF ARGUMENT C;
C     ORDER M,N; WITH N-M EVEN; UP TO ORDER N=M+2*NNMAX-2.
C     ALSO NORMALIZATION FUNCTION, N.
C
      SUBROUTINE OBRAD(C,M,IC,MMMAX,NNMAX,IRRMAX)
      COMMON EIG(50),D(50,50),DNEG(50),R1(50),F4(50)
      COMPLEX IX,R4,F4
      IX=(0.0,1.0)
      FFAC=1
      FAC=1
      FAC2=1
      GR0=1
      IF(M.EQ.0) GO TO 20
      MAXM=M+1
      DO 19 MM=2,MAXM
      IM=MM-1
      GR0=(2*IM-1)*(2*IM)*GR0
      FFAC=(2*IM-1)*FFAC
```

```
      FAC2=(2*IM-1)*(2*IM)*FAC2
19    FAC=IM*FAC
      IF(FFAC.GE.1.0E+17)GO TO 4
      IF(FAC2.GE.1.0E+30)GO TO 4
20    DO 17 NN=1,NNMAX
      N=2*(NN-1)+M
      SUM=0
      GR=GR0
      FNORM=0.
      DO 18 NR=1,IRRMAX
      IR=2*(NR-1)
      SUMP=GR*D(NN,NR)
      SUM=SUM+SUMP
      DMAG=ABS(D(NN,NR))
      IF(DMAG.LT.1.0E-30)GO TO 1
      FNORMP=2*GR*(D(NN,NR))**2/(2*IR+2*M+1)
2     FNORM=FNORM+FNORMP
      GRMAG=ABS(GR)
      IF(GRMAG.GT.1.0E+30) GO TO 5
      GR=(IR+2*M+1.)/(IR+1.)*(IR+2*M+2.)/(IR+2.)*GR
18    CONTINUE
      R1(NN)=(-1)**(NN-1)*2**M*FAC*C**M*D(NN,1)/((2*M+1)*SUM)
      R2=(-1)**(NN-1)*(2*M-1)*FAC*C**(M-1)/(2*FAC2)*3.14159
     1*(FFAC**2/SUM)*2**M/DNEG(NN)
      RR2=ABS(R2)
      IF(RR2.GE.1.0E+30)GO TO 4
      FFAC=(N+M+1)*FFAC/(N-M+2)
      IF(FFAC.GE.1.0E+17)NNMAX=NN
      R4=R1(NN)-IX*R2
      A5=ABS(FNORM)
      A1=ALOG10(A5)
      A4=ABS(R2)
      A2=ALOG10(A4)
      A3=A1+A2
      IF(A3.GT.30.)GO TO 3
      A5=ABS(R1(NN))
      A1=ALOG10(A5)
      A3=ABS(A1)+ABS(A2)
      IF(A3.GT.30.)GO TO 3
      F4(NN)=1/(FNORM*R4)
17    CONTINUE
      RETURN
1     FNORMP=0
      GO TO 2
3     F4(NN)=(0.,0.)
      GO TO 17
5     GR=0
      GO TO 18
4     MMMAX=M
      IW=1
      RETURN
      END
C
C     PSI FUNCTION
C
```

```
      SUBROUTINE FPSI(M,NNMAX)
      COMPLEX IX,F4,PSI
      COMMON EIG(50),D(50,50),DNEG(50),R1(50),F4(50)
     1,S0(50),P(50),SETA(50),SETA0(50),PSI(50)
      IX=(0.,1.)
      MM=M+1
      PSI(MM)=(0.,0.)
      DO 1 NN=1,NNMAX
      N=M+2*NN-2
      PSI(MM)=PSI(MM)+IX**(N)*S0(NN)**2*F4(NN)
    1 CONTINUE
      RETURN
      END
C
C     ASSOCIATED LEGENDRE POLYNOMIALS, P, OF ARGUMENT ETA0;
C     ORDER M,N; WITH N-M EVEN; UP TO ORDER N=M+2*NNMAX-2.
C
      SUBROUTINE POLYN(ETA0,M,NNMAX)
      DIMENSION PP(3)
      COMPLEX F4
      COMMON EIG(50),D(50,50),DNEG(50),R1(50),F4(50)
     1,S0(50),P(50)
      SQ=SQRT(1-ETA0*ETA0)
      PP(1)=0
      PP(2)=1
      IF(M.EQ.0)GO TO 1
      DO 2 L=1,M
    2 PP(2)=(2*L-1)*SQ*PP(2)
    1 P(1)=PP(2)
      DO 3 NN=2,NNMAX
      N=M+2*NN-3
      DO 4 L=1,2
      PP(3)=((2*N-1)*ETA0*PP(2)-(N+M-1)*PP(1))/(N-M)
      N=N+1
      PP(1)=PP(2)
    4 PP(2)=PP(3)
    3 P(NN)=PP(3)
      RETURN
      END
C
C     OBLATE SPHEROIDAL ANGULAR FUNCTIONS, S, OF ARGUMENTS
C     C AND ETA0; ORDER M,N; WITH N-M EVEN; UP TO ORDER
C     N=M+2*NNMAX-2.
C
      SUBROUTINE OBANG(NNMAX,IRRMAX)
      COMPLEX F4
      COMMON EIG(50),D(50,50),DNEG(50),R1(50),F4(50)
     1,S0(50),P(50),SETA(50),SETA0(50)
      DO 1 NN=1,NNMAX
      SETA(NN)=0
      DO 2 IRR=1,IRRMAX
    2 SETA(NN)=SETA(NN)+D(NN,IRR)*P(IRR)
    1 CONTINUE
      RETURN
      END
```

```
C
C     W FUNCTION
C
      SUBROUTINE FW(M,NNMAX)
      COMPLEX IX,F4,W,PSI
      COMMON EIG(50),D(50,50),DNEG(50),R1(50),F4(50)
     1,S0(50),P(50),SETA(50),SETA0(50),PSI(50),W(50)
      IX=(0.,1.)
      MM=M+1
      W(MM)=(0.,0.)
      DO 1 NN=1,NNMAX
      N=M+2*NN-2
    1 W(MM)=W(MM)+IX**N*SETA0(NN)*S0(NN)*F4(NN)
      RETURN
      END
C
C     Y FUNCTION
C
      SUBROUTINE FY(M,NNMAX)
      COMPLEX F4,Y0,YETA0,PSI,W
      COMMON EIG(50),D(50,50),DNEG(50),R1(50),F4(50)
     1,S0(50),P(50),SETA(50),SETA0(50),PSI(50),W(50),Y0(50)
     1,YETA0(50)
      MM=M+1
      Y0(MM)=(0.,0.)
      YETA0(MM)=(0.,0.)
      DO 1 NN=1,NNMAX
      N=M+2*NN-2
      Y0(MM)=Y0(MM)+(-1)**N*R1(NN)*S0(NN)*SETA(NN)*F4(NN)
    1 YETA0(MM)=YETA0(MM)+(-1)**N*R1(NN)*SETA0(NN)*SETA(NN)*F4(NN)
      RETURN
      END
C
C     X AND U FUNCTIONS
C
      SUBROUTINE FXU(MM,MMMAX)
      COMPLEX IX,F4,PSI,W,Y0,YETA0,U,X,PT
      COMMON EIG(50),D(50,50),DNEG(50),R1(50),F4(50)
     1,S0(50),P(50),SETA(50),SETA0(50),PSI(50),W(50),Y0(50)
     1,YETA0(50),U(50),X(50)
      IX=(0.,1.)
      IF(MM.EQ.1) GO TO 1
      IF(MM.EQ.MMMAX) GO TO 2
      PT=PSI(MM-1)*PSI(MM+1)
      PTT=CABS(PT)
      IF(PTT.EQ.0.)GO TO 3
      U(MM)=2*IX**(MM-2)*(W(MM-1)+W(MM+1))/(PSI(MM-1)+PSI(MM+1))
      X(MM)=2*IX**(MM-2)*(W(MM-1)-W(MM+1))/(PSI(MM-1)+PSI(MM+1))
      RETURN
    1 U(1)=-IX*W(2)/PSI(2)
      X(1)=(0.,0.)
      RETURN
    2 U(MM)=IX**(MM-2)*2*W(MM-1)/PSI(MM-1)
      X(MM)=U(MM)
      RETURN
```

```
3     MMMAX=MM-1
      RETURN
      END
C
C     Z FUNCTIONS
C
      SUBROUTINE FZ(MMMAX,Z,ZA,ZB,ZC,ZD)
      COMPLEX IX,IQ,Z,ZA,ZB,ZC,ZD,F4,PSI,W,Y0,YETA0,U,X
      COMMON EIG(50),D(50,50),DNEG(50),R1(50),F4(50)
     1,S0(50),R(50),SETA(50),SETA0(50),PSI(50),W(50),Y0(50)
     1,YETA0(50),U(50),X(50),CMPHI(50),SMPHI(50)
      IX=(0.,1.)
      Z=(0.,0.)
      ZA=Z
      ZB=Z
      ZC=Z
      ZD=Z
      DO 3 MM=1,MMMAX
      M=MM-1
      A=2
      B=1
      C=1
      IF(MM.EQ.1) A=1
      IF(MM.EQ.2) B=2
      IF(MM.EQ.2) C=0
      IQ=IX**(-M)
      Z=Z+A*YETA0(MM)*CMPHI(MM)
      IF(MM.EQ.1) GO TO 2
      ZA=ZA+IQ*(U(MM+1)*CMPHI(MM+1)-B*U(MM-1)*CMPHI(MM-1))
     1*Y0(MM)
      ZB=ZB+IQ*(U(MM+1)*SMPHI(MM+1)+U(MM-1)*SMPHI(MM-1))
     1*Y0(MM)
      ZC=ZC+IQ*(X(MM+1)*SMPHI(MM+1)-X(MM-1)*SMPHI(MM-1))
     1*Y0(MM)
      ZD=ZD+IQ*(X(MM+1)*CMPHI(MM+1)+C*X(MM-1)*CMPHI(MM-1))
     1*Y0(MM)
      GO TO 1
2     ZA=ZA+U(2)*CMPHI(2)*Y0(1)
      ZB=ZB+U(2)*SMPHI(2)*Y0(1)
      ZC=ZC+X(2)*SMPHI(2)*Y0(1)
      ZD=ZD+X(2)*CMPHI(2)*Y0(1)
1     CONTINUE
3     CONTINUE
      RETURN
      END
```

APPENDIX II
PROGRAM NOTES

The following notes are intended to clarify the program listed in Appendix I. The notation LN will be used to denote the line number in the listing, i.e., the first number on each line.

The first 9 lines of this program are non-executable and simply establish variable types and dimensions. LN10 through 69 accept input data and initialize variables. LN2, 20, 150 and 151 are associated with the interactive 'ESC' which permits interruption of the program and return to the start of the data input segment.

As noted in the body of the report, the scattered field is expressed as a triple summation over indices m, n, and r. In general, these indices range from 0, m, or 1 to some maximum value. In this program these indices are usually replaced by MM, NN, and IRR, respectively, where

$$MM = m+1$$
$$NN = (n-m)/2+1$$
$$IRR = 1,2,3,\ldots \quad .$$

Thus, for computational convenience, these indices all range upward from 1 in integer steps.

The truncations of these summations are determined on the basis of tests of key variables. During execution these variables are tested and upon reaching absolute values in the range 10^{30} to 10^{38}, the appropriate summation is truncated. The line numbers associated with these tests and the subsequent actions are tabulated below:

m	n	r
LN72	LN236	LN290
78	257-259	294
91	296-303	305-306
388	331-332	327-330
410-411	360-361	
434-436	364-365	
539-541	413	
551-552		

The necessary functions are formed by subroutine calls in Loop 10 and Loop 13 (LN70-94). These functions are then combined to form the scattered $\bar{E}$-fields and cross sections in LN95-128. LN130-132 provide the program output; and LN134-136 provide incrementing of the variable desired. LN138-149 contains the description of the coordinate system to be provided for an input of KA=-1. The remainder of the listing consists of the required function subroutines.

Whenever possible, variable names associated with the symbols presented in this report are used. One exception is the introduction of

$$F4 = \frac{1}{N_{mn}(-ic)R_{mn}^{(4)}(-ic;io)} .$$

Other exceptions are the Z functions appearing in LN95-98 and LN554-595. These are functions of U, X and Y found in Equation (9) of the text.